球面振り子の数値計算

<u>問題</u>

１．下図に示されるような球座標における各方向の加速度を導出せよ．

２．糸が伸びない球面振り子の運動方程式（r, θ, ϕ の 3 方向それぞれ）を表せ．

３．その数値解析法を考えよ．



<u>球座標系における加速度</u>

xy 平面上の単位ベクトルは，次式で与えられる．

$$\vec{l} = \cos\phi\vec{i} + \sin\phi\vec{j}$$

ここで，これを用いて，半径方向と子午線方向の単位ベクトルはそれぞれ

$$\vec{e}_r = \sin\theta\vec{l} + \cos\theta\vec{k}, \quad \vec{e}_\theta = \cos\theta\vec{l} - \sin\theta\vec{k}$$

となる．したがって，$\vec{l}$ を消去して，

$$\vec{e}_r = \sin\theta\cos\phi\vec{i} + \sin\theta\sin\phi\vec{j} + \cos\theta\vec{k},$$
$$\vec{e}_\theta = \cos\theta\cos\phi\vec{i} + \cos\theta\sin\phi\vec{j} - \sin\theta\vec{k}$$

また，経度方向の単位ベクトルは，

$$\vec{e}_\phi = -\sin\phi\vec{i} + \cos\phi\vec{j}$$

と与えられる．次に，時間で微分し，整理すると次式を得る．

$$\begin{aligned}\dot{\vec{e}}_r &= \frac{d}{dt}\left(\sin\theta\cos\phi\vec{i}\right)+\frac{d}{dt}\left(\sin\theta\sin\phi\vec{j}\right)+\frac{d}{dt}\left(\cos\theta\vec{k}\right)\\ &=\left(\dot\theta\cos\theta\cos\phi-\dot\phi\sin\theta\sin\phi\right)\vec{i}+\left(\dot\theta\cos\theta\sin\phi+\dot\phi\sin\theta\cos\phi\right)\vec{j}+\left(-\dot\theta\sin\theta\right)\vec{k}\\ &=\dot\theta\underbrace{\left(\cos\theta\cos\phi\vec{i}+\cos\theta\sin\phi\vec{j}-\sin\theta\vec{k}\right)}_{\vec{e}_\theta}+\dot\phi\sin\theta\underbrace{\left(-\sin\phi\vec{i}+\cos\phi\vec{j}\right)}_{\vec{e}_\phi}\end{aligned}$$

$$\begin{aligned}\dot{\vec{e}}_\theta &= -\dot\theta\sin\theta\cos\phi\vec{i}-\dot\phi\cos\theta\sin\phi\vec{i}-\dot\theta\sin\theta\sin\phi\vec{j}+\dot\phi\cos\theta\cos\phi\vec{j}-\dot\theta\cos\theta\vec{k}\\ &=-\dot\theta\underbrace{\left(\sin\theta\cos\phi\vec{i}+\sin\theta\sin\phi\vec{j}+\cos\theta\vec{k}\right)}_{\vec{e}_r}+\dot\phi\cos\theta\underbrace{\left(-\sin\phi\vec{i}+\cos\phi\vec{j}\right)}_{\vec{e}_\phi}\end{aligned}$$

$$\dot{\vec{e}}_\phi=-\dot\phi\cos\phi\vec{i}-\dot\phi\sin\phi\vec{j}=-\dot\phi\underbrace{\left(\cos\phi\vec{i}+\sin\phi\vec{j}\right)}_{\sin\theta\vec{e}_r+\cos\theta\vec{e}_\theta}$$

$$\left(\begin{aligned}\because\quad \vec{e}_r\sin\theta+\vec{e}_\theta\cos\theta &=\sin^2\theta\cos\phi\vec{i}+\sin^2\theta\sin\phi\vec{j}+\sin\theta\cos\theta\vec{k},\\ &\quad+\cos^2\theta\cos\phi\vec{i}+\cos^2\theta\sin\phi\vec{j}-\cos\theta\sin\theta\vec{k}=\cos\phi\vec{i}+\sin\phi\vec{j}\end{aligned}\right)$$

結局まとめると，

$$\dot{\vec{e}}_r=\dot\theta\vec{e}_\theta+\dot\phi\sin\theta\vec{e}_\phi,\quad \dot{\vec{e}}_\theta=-\dot\theta\vec{e}_r+\dot\phi\cos\theta\vec{e}_\phi,\quad \dot{\vec{e}}_\phi=-\dot\phi\left(\sin\theta\vec{e}_r+\cos\theta\vec{e}_\theta\right)$$

位置ベクトルは，

$$\vec{r}=r\vec{e}_r$$

と表され，球座標系の単位ベクトルが時間依存することに注意して，時間で微分すると，

$$\vec{v}=\dot{\vec{r}}=\dot r\vec{e}_r+r\dot{\vec{e}}_r=\dot r\vec{e}_r+r\dot\theta\vec{e}_\theta+r\dot\phi\sin\theta\vec{e}_\phi$$

さらに時間で微分し加速度は，

$$\begin{aligned}\ddot{\vec{r}}&=\ddot r\vec{e}_r+\dot r\dot{\vec{e}}_r+\dot r\dot\theta\vec{e}_\theta+r\ddot\theta\vec{e}_\theta+r\dot\theta\dot{\vec{e}}_\theta\\ &\quad+\dot r\dot\phi\sin\theta\vec{e}_\phi+r\ddot\phi\sin\theta\vec{e}_\phi+r\dot\phi\dot\theta\cos\theta\vec{e}_\phi+r\dot\phi\sin\theta\dot{\vec{e}}_\phi\end{aligned}$$

$$\begin{aligned}\ddot{\vec{r}}&=\ddot r\vec{e}_r+\dot r\dot\theta\vec{e}_\theta+\dot r\dot\phi\sin\theta\vec{e}_\phi+\dot r\dot\theta\vec{e}_\theta+r\ddot\theta\vec{e}_\theta-r\dot\theta^2\vec{e}_r+r\dot\theta\dot\phi\cos\theta\vec{e}_\phi\\ &\quad+\dot r\dot\phi\sin\theta\vec{e}_\phi+r\ddot\phi\sin\theta\vec{e}_\phi+r\dot\phi\dot\theta\cos\theta\vec{e}_\phi-r\dot\phi^2\sin^2\theta\vec{e}_r-r\dot\phi^2\sin\theta\cos\theta\vec{e}_\theta\end{aligned}$$

$$\begin{aligned}\ddot{\vec{r}}&=\left[\ddot r-r\dot\theta^2-r\dot\phi^2\sin^2\theta\right]\vec{e}_r+\left[r\ddot\theta+2\dot r\dot\theta-r\dot\phi^2\sin\theta\cos\theta\right]\vec{e}_\theta\\ &\quad+\left[r\ddot\phi\sin\theta+2\dot r\dot\phi\sin\theta+2r\dot\theta\dot\phi\cos\theta\right]\vec{e}_\phi\end{aligned}$$

糸が伸びないとき，$\ddot r=\dot r=0$ であるので，加速度は次式となる．

$$\vec{a}=\left[-r\dot\theta^2-r\dot\phi^2\sin^2\theta\right]\vec{e}_r+\left[r\ddot\theta-r\dot\phi^2\sin\theta\cos\theta\right]\vec{e}_\theta+\left[r\ddot\phi\sin\theta+2r\dot\theta\dot\phi\cos\theta\right]\vec{e}_\phi$$

運動方程式

重力加速度をg，糸の張力をT，長さをℓとして，運動方程式をたてる．下記の(1)-(3)式の左辺は，前問で導出した加速度項から得られる．右辺は，錘に作用する外力項（二次元の振り子と同様）からである．

＜r方向＞ $$\dot{\theta}^2+\dot{\phi}^2\sin^2\theta=-\frac{g}{\ell}\cos\theta+\frac{T}{m\ell} \qquad (1)$$

＜θ方向＞ $$\ddot{\theta}-\dot{\phi}^2\sin\theta\cos\theta=-\frac{g}{\ell}\sin\theta \qquad (2)$$

＜ϕ方向＞ $$\ddot{\phi}\sin\theta+2\dot{\theta}\dot{\phi}\cos\theta=0 \qquad (3)$$

最初の式は，動径方向（糸方向）の運動方程式であり，糸の張力Tを求めるためのものと考えられる．二番目の式は，振れ角θを求めるためのものと解釈され，三番目の式は，周方向の角度ϕを求めるためのものである．どの式にもθとϕが混ざっていて，数値積分するにもやや面倒に見える．そこで，(3)式を積分するために，両辺に$\sin\theta$をかける．

$$\ddot{\phi}\sin^2\theta+2\dot{\theta}\dot{\phi}\cos\theta\sin\theta=0$$

この式は，次式を時間tで微分した形になっているのがわかる．

$$\dot{\phi}\sin^2\theta=C_1 \qquad (4)$$

したがって，(3)式の代わりにこの式を用いる．この式の意味は，角運動量の保存則を表している．それを示すために，両辺に$m\ell^2$を掛けると，次式が得られる．

$$m\ell^2\dot{\phi}\sin^2\theta=m(\ell\sin\theta)^2\dot{\phi}=C_2$$

ここで$\ell\sin\theta$は，中心軸から錘までの距離を表す．この式の意味は，振れ角θが小さい時は，周方向の角速度$\dot{\phi}$が大きくなることを示している．また，その逆も成り立つ．周方向に与える初速度v_1は，以下の関係を持つ．

$$v_1=\ell\sin\theta_1\cdot\dot{\phi}_1 \qquad (5)$$

θ_1 は初期角度である．(4)は，$\dot{\phi}\sin^2\theta=C_1=\dot{\phi}_1\sin^2\theta_1$と考えても良いので，
(4)(5)の両式から，次式を得る．

$$\dot{\phi}=\frac{v_1\sin\theta_1}{\ell\sin^2\theta} \qquad (6)$$

これで，周方向の角速度$\dot{\phi}$がθの関数として与えられた．最後に(6)式を(2)式に代入し，次式を得る．

$$\dot{\omega}_\theta=-\frac{g}{\ell}\sin\theta+\left(\frac{v_1\sin\theta_1}{\ell}\right)^2\frac{\cos\theta}{\sin^3\theta},\quad \left(\because\quad \dot{\theta}\equiv\omega_\theta\right) \qquad (7)$$

この(7)式は，θのみになっていて，(2)式と比べると随分簡単になっている．結局，(2)式と(3)式を解く代わりに，(7)式と(6)式を解く．(1)式は，最初に述べたように，糸の張力Tを求めるために使う．錘の軌道を求めるだけなら，特に(1)式を計算する必要はない．

数値計算方法

計算手順としては，最初に(7)式を２回時間積分し，θを求める．次に(6)式を一度時間積分し，ϕを求める．最後に，球座標系における位置をカーテシアン座標系の位置に変換し，数値データとして収めれば良い．計算方法としては，いろいろ考えられるが，ここではLeap-Frog法を用いた場合を以下に示す．一回目の積分計算だけ，場合分けが必要である．結果については，研究室ホームページ Laboratory → Results → 「アニメーション」一番右奥にその可視化動画がありますので，そちらでご覧ください．

＜フォートランコード例＞

```
*==========================================================================
        PROGRAM spherical pendulum
*==========================================================================
*       ----- A-H と O-Z で始まる文字列は倍精度計算 ----------------
        IMPLICIT DOUBLE PRECISION (A-H,O-Z)
*       ----- 時間が 10000 ステップまで配列にする -----------------
        Dimension X(10000),Y(10000), Z(10000)
        Dimension THETA(10000), OMEGA(10000), FAI(10000)
        Dimension THETAN(10000), OMEGAN(10000), FAIN(10000)
*       ----- ファイルを開く-------------------------------------
          OPEN(30,FILE='result.txt')
*       ----- 重力加速度 ----------------------------------------
          G=9.8
*       ----- 円周率 -------------------------------------------
           PAI=3.14159265358
*       ----- 時間刻みの読み込み --------------------------------
           WRITE(*,*)'Time step?='
           READ(5,*) DELT
*       ----- 糸の長さの読み込み --------------------------------
           WRITE(*,*)'Length of the string?='
           READ(5,*) ZL
*       ----- 初期位置の読み込み --------------------------------
           WRITE(*,*)'Theta1 (Initial THETA angle in degree)?='
           READ(5,*) Theta1
          Theta1=Theta1*PAI/180.0
*       ----- φ方向の初速度の読み込み --------------------------
           WRITE(*,*)'V1 (Initial velocity in the FAI direction)?='
           READ(5,*) V1
*       ----- 計算の繰り返し回数の読み込み ----------------------
           WRITE(*,*)'How many steps do you compute?='
           READ(5,*) NN
*       ----- θ方向の初期角速度 --------------------------------
           OMEGA(1)=0.0
*       ----- 初期位置（θ方向）---------------------------------
           THETA(1)=Theta1
*       ----- 初期位置（φ方向）---------------------------------
          FAI(1)=0.0
*       ----- 初期位置のデカルト座標への変換 -----------------
           X(1)=ZL*sin(THETA(1))*cos(FAI(1))
          Y(1)=ZL*sin(THETA(1))*sin(FAI(1))
           Z(1)=ZL*(1.0-cos(THETA(1)))
*       ----- 初期位置のファイルへの書き出し ---------------------
          WRITE(30,112) X(1), Y(1), Z(1)
```

```
*=====================================================================
*================ 最初の計算の開始 ===================================
*=====================================================================
*        -----  一回目のθ方向の角速度の計算（時間刻みは半分）------
            OMEGAN(2)=OMEGA(1)+(-G/ZL*sin(THETA(1))
       &      +(V1/ZL)**2.0
       &         *(sin(Theta1))**2.0*cos(THETA(1))/sin(THETA(1))**3.0)
       &      *DELT/2.0
            OMEGA(2)=OMEGAN(2)
*        -----  一回目のθの計算  ---------------------------------
            THETAN(2)=THETA(1)+OMEGA(2)*DELT
            THETA(2)=THETAN(2)
*        -----  一回目のφの計算  ---------------------------------
           FAIN(2)=FAI(1)+(V1*sin(Theta1)/(ZL*sin(THETA(1))**2.0))*DELT
           FAI(2)=FAIN(2)
*        -----  一回目のデカルト座標への変換  -------------------
            X(2)=ZL*sin(THETA(2))*cos(FAI(2))
           Y(2)=ZL*sin(THETA(2))*sin(FAI(2))
            Z(2)=ZL*(1.0-cos(THETA(2)))
*=====================================================================
*================ 繰り返し計算の開始 =================================
*=====================================================================
            DO 20 I=2,NN-1
*        -----  θ方向の角速度の計算  ----------
            OMEGAN(I+1)=OMEGA(I)+(-G/ZL*sin(THETA(I))
       &      +(V1/ZL)**2.0
       &         *(sin(Theta1))**2.0*cos(THETA(I))/sin(THETA(I))**3.0)
       &      *DELT
            OMEGA(I+1)=OMEGAN(I+1)
*        -----  θの計算  -----------------------------------------
            THETAN(I+1)=THETA(I)+OMEGA(I+1)*DELT
            THETA(I+1)=THETAN(I+1)
*        -----  φの計算  -----------------------------------------
           FAIN(I+1)=FAI(I)+(V1*sin(Theta1)/(ZL*sin(THETA(I))**2.0))*DELT
           FAI(I+1)=FAIN(I+1)
*        -----  デカルト座標への変換  --------------------------
            X(I+1)=ZL*sin(THETA(I+1))*cos(FAI(I+1))
           Y(I+1)=ZL*sin(THETA(I+1))*sin(FAI(I+1))
            Z(I+1)=ZL*(1.0-cos(THETA(I+1)))
*        -----  ファイルへの書き出し  -----------------------------
         WRITE(30,112) X(I), Y(I), Z(I)
*        -----  画面への書き出し  ---------------------------------
         WRITE(6,113)I, X(I), Y(I), Z(I)
    20 CONTINUE
*        -----  フォーマット文  -----------------------------------
   112 FORMAT(3F12.5)
   113 FORMAT(I4, 3F12.5)
*=====================================================================
*================ 繰り返し計算の終了 =================================
*=====================================================================
         CLOSE(30)
         STOP
         END
```